An Exercise from Rotman (Carmichael) about Commutator Subgroups This question is inspired by a recent question by user Jaymes about the set of commutators not being a group.
Following a link to MO, Gerry Myerson posted an example from Carmichael. I quote
Let $G$ be a subgroup of $S_{16}$ generated by the following eight elements:
$$
\eqalign{(ac)(bd);&(eg)(fh);\cr(ik)(jl);&(mo)(np);\cr(ac)(eg)(ik);&(ab)(cd)(mo);\cr(ef)(gh)(mn)(op);&(ij)(kl).\cr}
$$
Then the commutator subgroup is generated by the first four elements above, and is of order $16$. Moreover,
$$
\alpha=(ik)(jl)(mo)(np)
$$
is in the commutator subgroup, but is not a commutator. 
However, no proof is given. Can somebody offer a proof of this? 
Regards,

I am not sure if this is a good enough of a proof for you, but I can work it out in Mathematica and confirm that with explicit computations. If this not helpful, I would be happy to delete the answer.

The copy and paste ready code follows:
Commutator[a_, b_] := 
 PermutationProduct[a, b, PermutationPower[a, -1], 
  PermutationPower[b, -1]]

ru = {"a" -> 1, "b" -> 2, "c" -> 3, "d" -> 4, "e" -> 5, "f" -> 6, 
   "g" -> 7, "h" -> 8, "i" -> 9, "j" -> 10, "k" -> 11, "l" -> 12, 
   "m" -> 13, "n" -> 14, "o" -> 15, "p" -> 16};

generators = {
   Cycles[{{1, 3}, {2, 4}}], Cycles[{{5, 7}, {6, 8}}],
   Cycles[{{9, 11}, {10, 12}}], Cycles[{{13, 15}, {14, 16}}],
   Cycles[{{1, 3}, {5, 7}, {9, 11}}], 
   Cycles[{{1, 2}, {3, 4}, {13, 15}}],
   Cycles[{{5, 6}, {7, 8}, {13, 14}, {15, 16}}], 
   Cycles[{{9, 10}, {11, 12}}]
   };

g = PermutationGroup[generators];

GroupOrder[g]

g1 = PermutationGroup[Take[generators, 4]];

GroupOrder[g1]

el = Cycles[{{9, 11}, {10, 12}, {13, 15}, {14, 16}}];

MemberQ[GroupElements[g1], el]

AllCommutators = 
 DeleteDuplicates[Commutator @@@ Tuples[GroupElements[g], {2}]]

Complement[GroupElements[g1], AllCommutators]

